
Unit 6: Drawing Skill Builder 3: Plotting a Point  

In this lesson, you will write a program to plot a point, 

automatically move it, and display its coordinates as it 

moves. Use [esc] to end the program. 

Objectives: 

 Plot a point  

 Move it automatically 

 Display its coordinates 

 Use getKey(0) to look for a keypress  

Plotting Points 

So far you have experienced drawing shapes (line, rectangle, circle and arc). There’s 

another tool available: PlotXY lets you plot a point using its coordinates (x, y). 

The syntax is: PlotXY x, y, style 

   x, y represent the coordinates of the point based on the current window setting.  

   style is a value from 1 to 13 and affects the size/shape of the point. It is required. 
 

Teacher Tip:  DrawText is introduced later in this lesson. DrawPoly and FillPoly 

statements are not addressed in this unit. Plot XY has 13 point styles:                                     

 

 

Plot a Point 

1. Start a new program named movepoint  

2. Set up a nice window: SetWindow -15, 15, -10, 10 

3. Begin by plotting the point at the origin:  

                    x:=0 

                    y:=0 

                    PlotXY x, y, 1 

PlotXY (found on [menu] >Draw > Shapes) plots the point (x,y) in style number 1. We 

use the variables x and y because they will change value in the program to make the 

point move. Try some of the other styles (1 to 13 for the third parameter).  

The Main Loop 

We will use a getkey(0) statement ([menu] > I/O) to look for the ‘esc’ key to end the 

program: 

4. After the PlotXY statement add: 

                   k:=” ”         (nothing between the quotes) 

                         While k≠”esc” 

                                                       (loop body) 

                             k:=getKey(0) 

                         EndWhile 

 

Teacher Tip:   

getKey(0) returns the latest keypress without pausing the program. In this instance the 

code above ends the loop (and the program) when [esc] is pressed.  

 

Move the Point 

The loop will make the point appear to move by changing the x-coordinate. But first, 

the point needs to be erased from its current position. We will ‘erase’ it by plotting it in 

the background color (white)… 

5. In the loop body between While and getKey(0) add: 

                   SetColor 255,255,255                 white 

                   PlotXY x, y, 1                              erase current point 

                   x:=x+1                                         increase x 

                   SetColor 0, 0, 0                          black 

                   PlotXY x, y, 1                             plot new point 

                   Wait 0.5                                      wait ½ second 

Wait is on [menu] > Control and tells the calculator to pause processing for 

that number of seconds. 

Run the program to see the point move to the right. Notice that it goes off the 

screen?  

Press [esc] to end the program and modify the program to keep the point 

with in the screen bounds: 

 

 

 

 

     

6. After x := x + 1 add the statement: 

                         if x > 15                         at the right edge of screen? 

                             x := -15                      go to the left edge of the screen 

 

Challenge: Can you make the point ‘turn around’ and move to the left when it reaches 

the right edge of the screen and move to the right when it reaches the left edge? 

Teacher Tip:  to ‘turn around’ at the edges of the screen:  

          Before the loop: 

              j := 1    (or any number… this is the increment value) 

          In the loop: 

             x := x + j                    (in place of x := x + 1) 

             If x > 15 or x < -15    (these values are the edges of the screen from SetWindow) 

                      j := - j 

Show the Coordinates 

While the drawing screen is showing, I/O statements like Disp and Request are 

disabled. To get text information to appear on the drawing screen we have to use 

DrawText (found on [menu] > Draw  > DrawText).  

             Syntax:     DrawText   thing1, thing2, thing3, … 

            things are: string literals like “Hello, world!” 

                                  numbers 

                                  variables 

                                  expressions 

            Example (without programming): on a Calculator app, type 

                    drawtext 30, 50, “The value of x is “, x 

                    (your value of x may differ!) 

 

 

7. Now, back to our program. Add some statements to display the coordinates 

of the point as it moves. Just before the Wait statement add the statements: 

                     SetColor 0, 0, 0                                    black 

                     DrawText -15, -10, “(“, x,” ,”, y, ”)”    display (x,y) 

 

 

 

 

Run the program now to see the coordinates written in the lower left corner of 

the screen. But the values are written over the previous ones, making them 

unreadable! 

8. The fix: erase the region where the coordinates are displayed using a filled 

white rectangle before drawing coordinates. Add: 

                     SetColor 255, 255, 255                        white 

                     FillRect -15, -10, 8, 3                            erase the coordinates 

just above:    SetColor 0, 0, 0                                     black 

                     DrawText -15, -10, “(“, x,” ,”, y, ”)”         display (x,y) 

Run the program again. Better?  

Add your own enhancements by changing the style and color of the point and 

the movement behavior. 

 

Teacher Tip:   

 The process of animation requires erasing the old image and drawing a new one. 

Sometimes this may require additional temporary variables to keep track of the ‘old’ 

data.  

 Displaying the coordinates using parentheses and a comma requires some delicate 

detail work with the quotes and the commas. Have patience! There are a lot of commas 

in that statement. 

 Clear -15,-10, 8, 3 has the same effect as the FillRect statement in this case: It clears 

the rectangular region.  

 The complete program:  

Define movepoint()= 

Prgm 

:  SetWindow −15, 15, −10, 10 

:  x := 0 

:  y := 0 

:  PlotXY x, y, 1 

:  k: = "" 

:  While k ≠ "esc" 

:  SetColor 255, 255, 255 

:  PlotXY x, y, 1 

:  x := x+1 

:  If x > 15 

:  x := −15 

:  SetColor 0, 0, 0 

:  PlotXY x, y, 1 

:  SetColor 255, 255, 255 

:  FillRect −15, −10, 6, 3 

:  SetColor 0, 0, 0 

:  DrawText −15, −10, "(",  x,", ", y, ")" 

:  Wait 0.5 

:  k := getKey(0) 

:EndWhile 

:EndPrgm 

 
 


